

Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills
Synopsys, Inc.
Maricopa, AZ
dillan@synopsys.com

Chip Haldane
The Chip Abides, LLC
Gilbert, AZ
chip@thechipabides.com

Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

I. CONSTRAINTS AND POLICY CLASS REVIEW

Random objects and constraints are the foundational building blocks of constrained random verification in SystemVerilog. The simplest implementations embed fixed constraints within a class definition. Embedded constraints lack flexibility; all randomized object instances must meet these requirements exactly as they are written.

In-line constraints using the **with** construct offer marginally better flexibility. Although these external constraints allow greater variability of random objects, their definitions are still fixed within the calling context. Furthermore, all in-line constraints must be specified within a single call to **randomize()**.

Policy classes are a technique for applying SystemVerilog constraints in a portable, reusable, and incremental manner, originally described by John Dickol [1] [2]. The operating mechanism leverages an aspect of "global constraints," the simultaneous solving of constraints across a set of random objects. Randomizing an object that contains policies also randomizes the policies. Meanwhile, the policies contain a reference back to the container. Consequently, the policy container is constrained by the policies it contains. Dickol's approach is illustrated by the following code.

```
1  class policy_base#(type ITEM=uvm_object);
2      ITEM item;
3
4      virtual function void set_item(ITEM item);
5          this.item = item;
6      endfunction
7  endclass
8
9  class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);
10     rand policy_base#(ITEM) policy[$];
11
12     function void add(policy_base#(ITEM) pcy);
13         policy.push_back(pcy);
14     endfunction
15
16     function void set_item(ITEM item);
17         foreach(policy[i]) policy[i].set_item(item);
18     endfunction
19 endclass
```

Fig. 1: The **policy_base** and **policy_list** classes

The first two classes provide the core definitions for policies. The **policy_base** implements the hook back to the policy container, and the **policy_list** enables related policies to be organized into groups. Both of these classes are parameterized by a container object type, so a unique specialization will be required for each policy container class that needs to be constrained.

Next, **addr_txn** is an example policy container—a generic transaction with a random address and size that implements policies.

```

1  class addr_txn;
2      rand bit [31:0] addr;
3      rand int      size;
4      rand policy_base#(addr_txn) policy[$];
5
6      constraint c_size {size inside {1, 2, 4};}
7
8      function void pre_randomize;
9          foreach(policy[i]) policy[i].set_item(this);
10     endfunction
11 endclass

```

Fig. 2: The **addr_txn** class

The next three classes implement some policies for the **addr_txn** class. A list of valid address ranges can be stored in the **ranges** array. The **addr_permit_policy** will choose one of the ranges at random and constrain the address to be within the range, while the **addr_prohibit_policy** will exclude the address from randomizing inside any of the ranges in its list.

```

1  class addr_policy_base extends policy_base#(addr_txn);
2      addr_range ranges[$];
3
4      function void add(addr_t min, addr_t max);
5          addr_range rng = new(min, max);
6          ranges.push_back(rng);
7      endfunction
8  endclass
9
10 class addr_permit_policy extends addr_policy_base;
11     rand int selection;
12
13     constraint c_addr_permit {
14         selection inside {[0:ranges.size()-1]};
15
16         foreach(ranges[i]) {
17             if(selection == i) {
18                 item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
19             }
20         }
21     }
22 endclass
23
24 class addr_prohibit_policy extends addr_policy_base;
25     constraint c_addr_prohibit {
26         foreach(ranges[i]) {
27             !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
28         }
29     }
30 endclass

```

```

29     }
30 endclass

```

Fig. 3: Policies for constraining the `addr_txn` class

The final class shows how policies might be used. The `addr_constrained_txn` class extends `addr_txn` and defines two policies, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The `addr_constrained_txn` class then passes the local `pcy` list to the parent `policy` queue.

```

1  class addr_constrained_txn extends addr_txn;
2      function new;
3          addr_permit_policy    permit = new;
4          addr_prohibit_policy  prohibit = new;
5          policy_list#(addr_txn) pcy = new;
6
7          permit.add('h00000000, 'h0000FFFF);
8          permit.add('h10000000, 'h1FFFFFFF);
9          pcy.add(permit);
10
11         prohibit.add('h13000000, 'h130FFFFF);
12         pcy.add(prohibit);
13
14         this.policy = {pcy};
15     endfunction
16 endclass

```

Fig. 4: The `addr_constrained_txn` subclass

At this point, an instance of `addr_constrained_txn` can be created and randomized like normal, and the address will be constrained based on the embedded policies.

```

1  addr_constrained_txn txn = new();
2  txn.randomize();

```

Fig. 5: Randomizing an instance of `addr_constrained_txn`

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [3]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [4]. Additionally, McClish defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing of the upgraded policy package implementation will also be discussed and resolved.

II. PROBLEM #1: PARAMETERIZED POLICIES

The first problem with the above policy implementation is that because policies are specialized using parameterization to the class they constrain, different specializations of parameterized classes cannot be grouped and indexed. This is most apparent when creating a class hierarchy. If you wanted to extend a class and add a new field to constrain then you need a new policy type. The new policy type requires an extra policy list to modify any new attributes in the child class and requires the user to know which layer of the class hierarchy defines each attribute they want to constrain. Imagine a complex class hierarchy with many layers of inheritance and extension, and new constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [5]). Using policies becomes cumbersome

in this case because you need to manage a separate policy list for each level of the hierarchy and know what each list can and cannot contain. For example, extending the `addr_txn` class to create a version with parity checking results in a class hierarchy that looks like this:

```

1  class addr_txn;
2      // ... unchanged from previous example
3  endclass
4
5  class addr_p_txn extends addr_txn;
6      rand bit parity;
7      rand bit parity_err;
8      rand policy_base#(addr_p_txn) addr_p_policy[$];
9
10     constraint c_parity_err {
11         soft (parity_err == 0);
12         (parity_err) ^ ($countones({addr, parity}) == 1);
13     }
14
15     function void pre_randomize;
16         super.pre_randomize();
17         foreach(addr_p_policy[i]) addr_p_policy[i].set_item(this);
18     endfunction
19 endclass
20
21 class addr_constrained_txn extends addr_p_txn;
22     function new;
23         addr_permit_policy      permit_p = new;
24         addr_prohibit_policy    prohibit_p = new;
25
26         // definition to follow in a later example - sets parity_err to the value
27         ↪ provided
28         addr_parity_err_policy  parity_err_p = new;
29
30         policy_list#(addr_txn)  addr_pcy_lst = new;
31         policy_list#(addr_p_txn) addr_p_pcy_lst = new;
32
33         permit_p.add('h00000000, 'h0000FFFF);
34         permit_p.add('h10000000, 'h1FFFFFFF);
35         addr_pcy_lst.add(permit_p);
36
37         prohibit_p.add('h13000000, 'h130FFFFF);
38         addr_pcy_lst.add(prohibit_p);
39
40         parity_err_p.set(1'b1);
41         addr_p_pcy_lst.add(parity_err_p);
42
43         this.addr_policy      = {addr_pcy_lst};
44         this.addr_p_policy    = {addr_p_pcy_lst};
45     endfunction
46 endclass

```

Fig. 6: Modified address transaction classes with a parity checking subclass and policies included

This class ends up with a lot of boilerplate code and is not very intuitive to use. The user needs to keep track of what the difference is between each policy list and apply constraints properly. Each additional subclass in a hierarchy will cause

the constrained class to grow further. The solution to this problem is to replace the parameterized policy base with a non-parameterized base and a parameterized extension. We chose to use an interface class as our non-parameterized container for the flexibility it offers over inheriting from a virtual base class—namely that our parameterized base is free to implement other interfaces, and should the need ever arise, other classes might implement the policy interface.

```

1 interface class policy;
2     pure virtual function void set_item(uvm_object item);
3 endclass
4
5 virtual class policy_imp#(type ITEM=uvm_object) implements policy;
6     protected rand ITEM m_item;
7
8     virtual function void set_item(uvm_object item);
9         if (!$cast(m_item, item)) begin
10             `uvm_warning("policy::set_item()", "Item/policy type mismatch")
11             this.m_item = null;
12             this.m_item.rand_mode(0);
13         end
14     endfunction: set_item
15 endclass: policy_imp
16
17 typedef policy policy_queue[$];

```

Fig. 7: The **policy** interface class and **policy_imp** class

Implementing a non-parameterized base (in this case, an interface class) enables all policies targeting a particular class hierarchy to be stored within a single common **policy_queue** implemented by the base class of the target class hierarchy. The **uvm_object** class is a good base class when using policies within UVM, as it will likely cover every class you will be randomizing. If working with another framework, a reasonable base class will need to be chosen. Because the **set_item** function no longer receives a parameterized type input, the input needs to be cast to the implementation type. If this cast fails, some form of cleanup is necessary. In the example above, the item is set to **null** and randomization is disabled when the cast fails.

```

1 class addr_policy extends policy_imp#(addr_txn);
2     // ... unchanged from previous example, with updated class extension
3 endclass
4
5 class addr_parity_err_policy extends policy_imp#(addr_p_txn);
6     protected bit parity_err;
7
8     constraint c_fixed_value {m_item != null -> m_item.parity_err == parity_err;}
9
10    function new(int parity_err);
11        this.parity_err = parity_err;
12    endfunction
13 endclass
14
15 class addr_permit_policy extends addr_policy;
16     // same as before
17 endclass
18
19 class addr_prohibit_policy extends addr_policy;
20     // same as before

```

```
21 endclass
```

Fig. 8: Address policies updated to use the new policy implementation

The address policies are updated to inherit from **policy_imp** parameterized to the relevant transaction class. Additionally, the constraints are updated to verify the internal item handle is not null.

```
1  class addr_txn extends uvm_object;
2      rand policy_queue policy;
3      // policy is replaced with the above. All other members, constraints,
4      // and pre_randomize are unchanged from the previous example
5  endclass
6
7  class addr_p_txn extends addr_txn;
8      rand bit parity;
9      rand bit parity_err;
10     constraint c_parity_err {/*....*/}
11     // The local addr_p_policy and pre_randomize are removed. Everything
12     // else is unchanged from the previous example
13 endclass
14
15 class addr_constrained_txn extends addr_p_txn;
16     function new;
17         // only a single policy queue is necessary now
18         policy_queue pcy;
19
20         addr_permit_policy    permit_p    = new();
21         addr_prohibit_policy  prohibit_p  = new();
22         addr_parity_err_policy parity_err_p;
23
24         permit_p.add('h00000000, 'h0000FFFF);
25         permit_p.add('h10000000, 'h1FFFFFFF);
26         pcy.push_back(permit_p);
27
28         prohibit_p.add('h13000000, 'h130FFFFF);
29         pcy.push_back(prohibit_p);
30
31         parity_err_p = new(1'b1);
32         pcy.push_back(parity_err_p);
33
34         this.policy = {pcy};
35     endfunction
36 endclass
```

Fig. 9: Address transaction classes updated to use the new policy implementation

The new address transaction classes are simplified considerably. Only a single policy queue is required in the class hierarchy, and the **addr_constrained_txn** class does not need to keep track of any specialized lists. The **addr_txn** class now subclasses **uvm_object** to provide compatibility with the policy interface.

III. PROBLEM #2: DEFINITION LOCATION

The second problem with policies is “where do I define my policy classes?” This is not a complicated problem to solve, and most users likely place their policy classes in a file close to the class they are constraining. However, by explicitly pairing policy definitions with the class they constrain, users never need to hunt for the definitions or guess where they are located, and the definitions are given an appropriate and consistent home in the environment. This is done by embedding a **POLICIES**

class inside the class being constrained. This class acts as a container for all relevant policies and allows policies to be created by simply referencing the embedded class through the `::` operator.

```

1  class addr_txn extends uvm_object;
2      // class members, constraints, and pre_randomize unchanged from previous
   ↪ example
3
4      class POLICIES;
5          class addr_policy extends policy_imp#(addr_txn);
6              // ... unchanged from previous standalone class example
7          endclass
8
9          class addr_permit_policy extends addr_policy;
10             // ... unchanged from previous standalone class example
11         endclass
12
13         class addr_prohibit_p_policy extends addr_policy;
14             // ... unchanged from previous standalone class example
15         endclass
16     endclass: POLICIES
17 endclass
18
19 class addr_p_txn extends addr_txn;
20     // class members and constraints unchanged from previous example
21
22     class POLICIES extends addr_txn::POLICIES;
23         class addr_parity_err_policy extends policy_imp#(addr_p_txn);
24             // ... unchanged from previous example
25         endclass
26
27         static function addr_parity_err_policy PARITY_ERR(bit value);
28             PARITY_ERR = new(value);
29         endfunction
30     endclass: POLICIES
31 endclass
32
33 class addr_constrained_txn extends addr_p_txn;
34     function new;
35         policy_queue pcy;
36
37         addr_constrained_txn::POLICIES::addr_permit_policy  permit_p  = new();
38         addr_constrained_txn::POLICIES::addr_prohibit_policy prohibit_p = new();
39
40         // policy constraint value setup unchanged from previous example
41
42         pcy.push_back(addr_constrained_txn::POLICIES::PARITY_ERR(1'b1));
43
44         this.policy = {pcy};
45     endfunction
46 endclass

```

Fig. 10: Address transaction classes with embedded policies

In the above code, there were few changes from the previous iteration. All the policy classes were migrated into an embedded **POLICIES** class, and a static constructor was created for the parity error policy to simplify its creation. The

addr_p_txn::POLICIES class was set as a child of the **addr_txn::POLICIES** class, which further simplified the usage of the classes in the constraint block. Finally, the **addr_constrained_txn::POLICIES::** accessor is used to access policies in the constraint block.

A nice side effect of embedding policy class definitions within their target classes is that policies have access to protected members of the target class. This enables meta members of the target class (e.g., "knobs" such as **parity_err**) to be defined as protected, preventing external classes from manipulating them directly. In fact, a more advanced use of policies might define all members of a target class as protected, restricting the setting of fields exclusively through policies and reading through accessor functions, thus encouraging maximum encapsulation/loose coupling, which reduces the cost of maintaining and enhancing code and prevents bugs from cascading into classes that use policy-enabled classes.

IV. PROBLEM #3: BOILERPLATE OVERLOAD

The third problem with using policies is that policies are relatively expensive to define since you need at a minimum: a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints (property equals X), range constraints (property between Y and Z), or set membership (keyword **inside**) constraints, macros can be used to drastically reduce the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded **POLICIES** class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for complex constraints, if necessary.

```

1  // Fixed-value policy class and constructor macro
2  `define fixed_policy(POLICY, FIELD, TYPE) \
3  `m_fixed_policy_class(POLICY, FIELD, TYPE) \
4  `m_fixed_policy_constructor(POLICY, TYPE)
5
6  `define m_fixed_policy_class(POLICY, FIELD, TYPE) \
7      class POLICY`_policy extends base_policy; \
8          protected TYPE l_val; \
9
10         constraint c_fixed_value { \
11             (m_item != null) -> (m_item.FIELD == TYPE'(l_val)); \
12         } \
13
14         function new(TYPE value); \
15             this.l_val = value; \
16         endfunction \
17     endclass: POLICY`_policy
18
19 `define m_fixed_policy_constructor(POLICY, TYPE) \
20     static function POLICY`_policy POLICY(TYPE value); \
21         POLICY = new(value); \
22     endfunction: POLICY

```

Fig. 11: Macros for setting up the embedded **POLICIES** class and a fixed value policy

This example includes a **`fixed_policy** macro, which wraps two additional macros responsible for creating a policy class and a static constructor for the class. This **`fixed_policy** example policy class lets you constrain a property to a fixed value. A more complete macro definition can be found in Appendix B. The appendix includes **`start_policies**, **`start_extended_policies**, and **`end_policies** macros that are used to create the embedded **POLICIES** class within the constrained class instead of using hard-coded **class** and **endclass** statements. They set up class inheritance as needed and create a local typedef for the **policy_imp** parameterized type.

```

1  class addr_txn extends uvm_object;

```



```

2      // class members, constraints, and pre_randomize unchanged from previous
↪      example
3
4      `start_policies(addr_txn)
5          `include "addr_policies.svh"
6      `end_policies
7  endclass
8
9  class addr_p_txn extends addr_txn;
10     // class members and constraints unchanged from previous example
11
12     `start_extended_policies(addr_p_txn, addr_txn)
13         `fixed_policy(PARITY_ERR, bit, parity_err)
14     `end_policies
15 endclass
16
17 class addr_constrained_txn extends addr_p_txn;
18     // ... unchanged from previous example
19 endclass

```

Fig. 12: Simplified address transaction classes using the policy macros

The base `addr_txn` class has complex policies with a relationship between the `addr` and `size` fields, so rather than creating a policy macro that will only be used once, they can either be left as-is within the embedded policies class, or moved to a separate file and included with ``include` as was done here to keep the transaction class simple. The child parity transaction class is able to use the ``fixed_policy` macro to constrain the `parity_err` field. The constraint block remains the same as the previous example.

V. PROBLEM #4: UNEXPECTED POLICY REUSE BEHAVIOR AND OPTIMIZING FOR LIGHTWEIGHT POLICIES

A fourth problem we noticed in our initial usage of policy classes was occasional unexpected behavior when attempting to reuse a policy. It is hard to quantify and we never spent time to create a reproducible test case, but some of the policies behaved like they “remembered” being randomized when used repeatedly and wouldn’t reapply their constraints during subsequent `randomize` calls. Rather than spending time to diagnose and fix this issue, we were focusing on transitioning our policy approach to a “use once and discard” approach, where we would treat a policy as disposable and apply fresh policy instances before re-randomizing a target object. This improvement happened to resolve the randomization issue as well, so we never had a reason to circle back.

Fortunately, creating new policy instances for each randomization is not prohibitively expensive. We anticipated that we would need hundreds or thousands of policies per test, so we aimed to minimize the memory requirements and streamline functionality as much as possible. To minimize the footprint of policies, we chose to implement them as simple classes rather than extensions of `uvm_object` (the policy container itself still inherits from `uvm_object`). We duplicated some useful capabilities such as a basic copy routine, while omitting more expensive functionality such as factory compatibility or object comparison. The buggy randomization behavior implied we needed to pass in fresh policy instances each time we randomized an object. Two features helped to accomplish this: static constructor functions and a `copy` method.

```

1  static function addr_parity_err_policy PARITY_ERR(bit value);
2      PARITY_ERR = new(value);
3  endfunction
4  ...
5  pcy.push_back(addr_constrained_txn::POLICIES::PARITY_ERR(1'b1));

```

Fig. 13: Example static constructor function from the `PARITY_ERR` policy class

Using static functions to create new instances of the policies provides an easy way to create and use a policy with a single function call. The policy is created and immediately pushed onto the policy queue, then discarded and recreated the next time it is needed.

```

1 interface class policy;
2     ...
3     pure virtual function policy copy();
4 endclass: policy

```

Fig. 14: The **copy** method in the **policy** interface class

Adding a **copy** method to the policy classes provides a way for a policy to return a fresh copy of itself with the same configuration. This is useful when a policy needs to be reused multiple times. The policy can be copied and pushed onto the policy queue multiple times, or onto other policy queues, and each copy will constrain the field separately. The **copy** method is **pure virtual**, so it must be implemented by each policy class.

After migrating to this pattern of reuse, the issues we were seeing with unapplied constraints disappeared.

VI. MORE IMPROVEMENTS TO THE POLICY PACKAGE

The examples presented so far are functional, but are lacking many features that would be useful in a real-world implementation. The following examples will present additional improvements to the policy package that will make it more practical and efficient to use.

A. Expanding the **policy** interface class

The following **policy** interface class adds additional methods for managing a policy.

```

1 interface class policy;
2
3     pure virtual function string name();
4     pure virtual function string type_name();
5     pure virtual function string description();
6     pure virtual function bit item_is_compatible(uvm_object item);
7     pure virtual function void set_item(uvm_object item);
8     pure virtual function policy copy();
9
10 endclass: policy
11
12 `endif

```

Fig. 15: Expanded **policy** interface class

The **name**, **description**, and **copy** methods are implemented by the policy (or policy macro) directly and provide reporting information useful when printing messages about the policy to the log for the former two, or specific behavior for making a copy for the latter. The remaining three methods are implemented by **policy_imp** and are shared by all policies.

B. Better type safety checking and reporting in **policy_imp** methods

Some of the benefits of above methods can be seen by examining the new **set_item** method used by **policy_imp**.

```

1 virtual function void set_item(uvm_object item);
2     if (item == null) begin
3         `uvm_error("policy::set_item()", "NULL item passed")
4
5     end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
6         `uvm_info(
7             "policy::set_item()",
8             $sformatf(
9                 "policy <%s> applied to item <%s>: %s",
10                 this.name(), item.get_name(), this.description()

```

```

11         ),
12         UVM_FULL
13     )
14     this.m_item.rand_mode( 1 );
15
16     end else begin
17         `uvm_warning(
18             "policy::set_item()",
19             $sformatf(
20                 "Item <%s> type <%s> is not compatible with policy <%s> type <%s>",
21                 item.get_name(), item.get_type_name(), this.name(), this.type_name()
22             )
23         )
24         this.m_item = null;
25         this.m_item.rand_mode( 0 );
26     end
27 endfunction: set_item

```

Fig. 16: **set_item** method from **policy_imp**

The **set_item** method makes use of all the reporting methods to provide detailed log messages when **set_item** succeeds or fails. Additionally, the **item_is_compatible** is used before the **\\$cast** method is called and the **rand_mode** state is kept consistent with the result of the cast.

C. Replacing **policy_list** with **policy_queue**

Eagle-eyed readers might have noticed the lack of presence of a **policy_list** class in any of the examples above after migrating to the improved policy interface. Rather, a single typedef is all that is necessary to manage policies in a class.

```

1 typedef policy policy_queue[$];

```

Fig. 17: The **policy_queue** typedef

The **policy_queue** type is capable of storing any policy that implements the **policy** interface. The default queue methods are sufficient for aggregating policies, and in practice we found that using policy queues as containers was more efficient than **policy_list** instances. For example, for functions expecting a **policy_queue** argument we can directly pass in array literals populated by calls to static constructor functions, allowing us to define, initialize, aggregate, and pass policies all in a single line of code!

D. Provide a common implementation for policy-enabled classes using the **policy_container** interface and **policy_object** mixin

Our solution encourages loosely coupled code by defining APIs that pass **policy_queue** arguments and hide a target object's policies queue from other classes.

```

1 interface class policy_container;
2
3     // Queries
4     pure virtual function bit has_policies();
5
6     // Assignments
7     pure virtual function void set_policies(policy_queue policies);
8     pure virtual function void add_policies(policy_queue policies);
9     pure virtual function void clear_policies();
10
11     // Access

```

```

12     pure virtual function policy_queue get_policies();
13
14     // Copy
15     pure virtual function policy_queue copy_policies();
16
17 endclass: policy_container
18
19 `endif

```

Fig. 18: The **policy_container** interface class

This interface class should be implemented by any class that needs to use policies. The method implementations can then be used to manipulate a policy queue as needed, including adding to, replacing, or clearing the queue, and returning a handle to the queue or a copy of the queue to use elsewhere.

The **policy_container** interface class allows for a base class implementation that can be used to manage constraints across the entire class hierarchy.

```

1 class policy_object #(type BASE=uvmm_object) extends BASE implements
  ↳ policy_container;
2
3     protected policy_queue m_policies;
4
5     // Queries
6     virtual function bit has_policies();
7         // returns true/false based on size of m_policies
8     endfunction: has_policies
9
10    // Assignments
11    virtual function void set_policies(policy_queue policies);
12        // sets m_policies to a new queue of policies
13    endfunction: set_policies
14
15    virtual function void add_policies(policy_queue policies);
16        // adds new policies to m_policies
17    endfunction: add_policies
18
19    virtual function void clear_policies();
20        // clears m_policies
21    endfunction: clear_policies
22
23    // Access
24    virtual function policy_queue get_policies();
25        // return a handle to m_policies
26    endfunction: get_policies
27
28    // Copy
29    virtual function policy_queue copy_policies();
30        // a copy of m_policies
31    endfunction: copy_policies
32 endclass: policy_object

```

Fig. 19: A **policy_object** base class implementation

This base class implements all the **policy_container** functionality (a complete example is available in Appendix A). Packaging a common implementation and interface into a mixin enables sharing a common implementation and API for policy

support among any classes that might benefit from the use of policies, as seen in the following example.

```
1 // Use policy_object for transactions
2 class base_txn extends policy_object #(uvm_sequence_item);
3
4 // Use policy_object for sequences
5 class base_seq #(type REQ=uvm_sequence_item, RSP=REQ) extends policy_object #(
6   ↪ uvm_sequence #(REQ, RSP) );
7
8 // Use policy_object for configuration objects
9 class cfg_object extends policy_object #(uvm_object);
```

Fig. 20: Example classes using the **policy_object** base class

E. Protecting the policy queue enforces loosely coupled code

A subtle but significant additional benefit to using a base **policy_object** class along with the **policy_container** API is the ability to mark the container's policy queue as protected and prevent direct access to it.

The original implementation called **set_item** on each policy during the **pre_randomize** stage. That was necessary because the **policy_queue** was public, so there was nothing to prevent callers from adding policies without linking them to the target class.

Using an interface class and making the implementation private means the callers may only set or add policies using the available interface class routines, and because those routines are solely responsible for applying policies, they can also check compatibility (and filter incompatible policies) and call **set_item** immediately. This can be seen in the example **policy_object** implementation above, in the usage of the protected function **try_add_policy**.

By calling **set_item** when the policy is added to the queue, all policies will be associated with the target item automatically, so there is no need to do it during **pre_randomize**. This eliminates an easily-overlooked requirement for classes extending **policy_object** to make sure they call **super.pre_randomize()** in their local **pre_randomize** function.

VII. CONCLUSION

The improvements to the policy package presented in this paper provide a more robust and efficient implementation of policy-based constraints for SystemVerilog. The policy package is now capable of managing constraints across an entire class hierarchy, and the policy definitions are tightly paired with the class they constrain. The use of macros reduces the expense of defining common policies, while still allowing great flexibility in any custom policies necessary.

The complete policy package is available in Appendix A, with additional macro examples available in Appendix B. A functional package is also available for download [6] which can be included directly in a project to start using policies immediately.

REFERENCES

- [1] J. Dickol, "SystemVerilog Constraint Layering via Reusable Randomization Policy Classes," DVCon, 2015.
- [2] J. Dickol, "Complex Constraints: Unleashing the Power of the VCS Constraint Solver," SNUG Austin, September 29, 2016.
- [3] K. Vasconcellos, J. McNeal, "Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution," DVCon, 2021.
- [4] C. McClish, "Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins," DVCon, 2021.
- [5] J. Vance, J. Montesano, M. Litterick, J. Sprott, "Be a Sequence Pro to Avoid Bad Con Sequences," DVCon, 2019.
- [6] D. Mills, C. Haldane, "policy_pkg Source Code," https://github.com/DillanCMills/policy_pkg.

APPENDIX A POLICY PACKAGE

A. *policy_pkg.sv*

```
1  `ifndef __POLICY_PKG__
2  `define __POLICY_PKG__
3
4  `include "uvm_macros.svh"
5
6  package policy_pkg;
7
8      import uvm_pkg::*;
9
10     `include "policy.svh"
11
12     typedef policy policy_queue[$];
13
14     `include "policy_imp.svh"
15
16     `include "policy_container.svh"
17     `include "policy_object.svh"
18
19 endpackage: policy_pkg
20
21 `endif
```

B. *policy.svh*

```
1  `ifndef __POLICY__
2  `define __POLICY__
3
4  interface class policy;
5
6      pure virtual function string name();
7      pure virtual function string type_name();
8      pure virtual function string description();
9      pure virtual function bit item_is_compatible(uvm_object item);
10     pure virtual function void set_item(uvm_object item);
11     pure virtual function policy copy();
12
13 endclass: policy
14
15 `endif
```

C. *policy_imp.svh*

```
1  `ifndef __POLICY_IMP__
2  `define __POLICY_IMP__
3
4  virtual class policy_imp #(type ITEM=uvm_object) implements policy;
5
6      protected rand ITEM m_item;
```

```

7
8     pure virtual function string name();
9     pure virtual function string description();
10    pure virtual function policy copy();
11
12    virtual function string type_name();
13        return( ITEM::type_name );
14    endfunction: type_name
15
16    virtual function bit item_is_compatible(uvm_object item);
17        ITEM local_item;
18
19        return( (item != null) && ($cast(local_item, item)) );
20    endfunction: item_is_compatible
21
22    virtual function void set_item(uvm_object item);
23        if (item == null) begin
24            `uvm_error("policy::set_item()", "NULL item passed")
25
26        end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item))
↪    begin
27            `uvm_info(
28                "policy::set_item()",
29                $sformatf(
30                    "policy <%s> applied to item <%s>: %s",
31                    this.name(), item.get_name(), this.description()
32                ),
33                UVM_FULL
34            )
35            this.m_item.rand_mode( 1 );
36
37        end else begin
38            `uvm_warning(
39                "policy::set_item()",
40                $sformatf(
41                    "Cannot apply policy '%0s' of type '%0s' to target object '%0s'
↪    of incompatible type '%0s'",
42                    this.name(), ITEM::type_name(), item.get_name(),
↪    item.get_type_name()
43                )
44            )
45            this.m_item = null;
46            this.m_item.rand_mode( 0 );
47        end
48    endfunction: set_item
49
50 endclass: policy_imp
51
52 `endif

```

D. policy_container.svh

```

1 `ifndef __POLICY_CONTAINER__
2 `define __POLICY_CONTAINER__

```

```

3
4 interface class policy_container;
5
6     // Queries
7     pure virtual function bit has_policies();
8
9     // Assignments
10    pure virtual function void set_policies(policy_queue policies);
11    pure virtual function void add_policies(policy_queue policies);
12    pure virtual function void clear_policies();
13
14    // Access
15    pure virtual function policy_queue get_policies();
16
17    // Copy
18    pure virtual function policy_queue copy_policies();
19
20 endclass: policy_container
21
22 `endif

```

E. policy_object.svh

```

1 `ifndef __POLICE_OBJECT__
2 `define __POLICE_OBJECT__
3
4 class policy_object #(type BASE=uvm_object) extends BASE implements
↳ policy_container;
5
6     protected policy_queue m_policies;
7
8     // Queries
9     virtual function bit has_policies();
10    return( this.m_policies.size() > 0 );
11    endfunction: has_policies
12
13    // Assignments
14    virtual function void set_policies(policy_queue policies);
15    if( this.has_policies() ) begin
16        `uvm_warning( "policy", "set_policies() replacing existing policies" )
17    end
18    this.m_policies = {};
19    foreach( policies(i) ) try_add_policy( policies[i] );
20    endfunction: set_policies
21
22    virtual function void add_policies(policy_queue policies);
23    foreach( policies(i) ) try_add_policy( policies[i] );
24    endfunction: add_policies
25
26    virtual function void clear_policies();
27    if ( this.has_policies() ) begin
28        `uvm_info( "policy", $sformatf("clearing [%0d] policies from %s",
↳ this.m_policies.size(), this.get_name()), UVM_FULL)
29    end

```



```

30     this.m_policies = {};
31 endfunction: clear_policies
32
33 // Access
34 virtual function policy_queue get_policies();
35     return( this.m_policies );
36 endfunction: get_policies
37
38 // Copy
39 virtual function policy_queue copy_policies();
40     copy_policies = {};
41     foreach( this.m_policies[i] ) begin
42         copy_policies.push_back( this.m_policies[i].copy() );
43     end
44 endfunction: copy_policies
45
46 protected function void try_add_policy( policy new_policy );
47     if (new_policy.item_is_compatible(this)) begin
48         `uvm_info( "policy", $sformatf("adding policy %s to %s",
↵ new_policy.name, this.get_name()), UVM_FULL)
49         new_policy.set_item( this );
50         this.m_policies.push_back( new_policy );
51     end else begin
52         `uvm_warning( "policy", $sformatf("policy %s not compatible with target
↵ %s", new_policy.name, this.get_name()))
53     end
54 endfunction: try_add_policy
55 endclass: policy_object
56
57 `endif

```

APPENDIX B POLICY MACROS

A. *policy_macros.svh*

```

1  `ifndef __POLICY_MACROS__
2  `define __POLICY_MACROS__
3
4  //=====
5  // Embedded POLICIES class macros
6  //=====
7
8  `define start_policies(cls)                                \
9  class POLICIES;                                           \
10 `m_base_policy(cls)
11
12 `define start_extended_policies(cls, parent)              \
13 class POLICIES extends parent::POLICIES;                  \
14 `m_base_policy(cls)
15
16 `define end_policies                                     \
17 endclass: POLICIES
18
19 `define m_base_policy(cls)                                \
20     typedef policy_imp#cls) base_policy;
21
22
23 //=====
24 // Policy template macros
25 //=====
26
27 `include "constant_policy.svh"
28 `include "fixed_policy.svh"
29 `include "member_policy.svh"
30 `include "range_policy.svh"
31
32 `endif

```

B. *constant_policy.svh*

```

1  `ifndef __CONSTANT_POLICY__
2  `define __CONSTANT_POLICY__
3
4  // Full policy definition
5  `define constant_policy(POLICY, FIELD, TYPE, CONST)      \
6  `m_const_policy_class(POLICY, FIELD, TYPE, CONST)        \
7  `m_const_policy_constructor(POLICY)
8
9  // Policy class definition
10 `define m_const_policy_class(POLICY, FIELD, TYPE, CONST) \
11     class POLICY`_policy extends base_policy              \
12
13         constraint c_policy_constraint {                   \
14             (m_item != null) -> (m_item.FIELD == TYPE'(CONST));

```

```

15         }
16
17         function new();
18         endfunction: new
19
20         virtual function string name();
21             return ("POLICY");
22         endfunction: name
23
24         virtual function string description();
25             return ("(FIELD==CONST)");
26         endfunction: description
27
28         virtual function policy copy();
29             copy = new();
30         endfunction: copy
31     endclass: POLICY`__policy
32
33 // Policy constructor definition
34 `define m_const_policy_constructor(POLICY)
35     static function POLICY`__policy POLICY();
36         POLICY = new();
37     endfunction: POLICY
38
39 `endif

```

C. fixed_policy.svh

```

1  `ifndef __FIXED_POLICY__
2  `define __FIXED_POLICY__
3
4  // Full policy definition
5  `define fixed_policy(POLICY, FIELD, TYPE, RADIX="%0p")
6  `m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX)
7  `m_fixed_policy_constructor(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
11     class POLICY`__policy extends base_policy
12         local TYPE          l_val;
13         local string        l_radix=RADIX;
14
15         constraint c_policy_constraint {
16             (m_item != null) -> (m_item.FIELD == TYPE'(l_val));
17         }
18
19         function new(TYPE value, string radix=RADIX);
20             this.set_value(value);
21             this.set_radix(radix);
22         endfunction: new
23
24         virtual function string name();
25             return ("POLICY");
26         endfunction: name

```

```

27         virtual function string description();
28         return ({
29             ~(FIELD==",
30             $sformatf(l_radix, l_val),
31             ~")~"
32         });
33     endfunction: description
34
35     virtual function policy copy();
36     copy = new(l_val, l_radix);
37     endfunction: copy
38
39     virtual function void set_value(TYPE value);
40     this.l_val = value;
41     endfunction: set_value
42
43     virtual function TYPE get_value();
44     return (this.l_val);
45     endfunction: get_value
46
47     virtual function void set_radix(string radix);
48     this.l_radix = radix;
49     endfunction: set_radix
50
51     virtual function string get_radix();
52     return (this.l_radix);
53     endfunction: get_radix
54 endclass: POLICY` `_policy
55
56 // Policy constructor definition
57 `define m_fixed_policy_constructor(POLICY, TYPE, RADIX="%0p")
58     static function POLICY` `_policy POLICY(TYPE value, string radix=RADIX);
59     POLICY = new(value, radix);
60     endfunction: POLICY
61
62 `endif
63

```

D. member_policy.svh

```

1  `ifndef __MEMBER_POLICY__
2  `define __MEMBER_POLICY__
3
4  // Full policy definition
5  `define member_policy(POLICY, FIELD, TYPE, RADIX="%0p")
6  `m_member_policy_class(POLICY, FIELD, TYPE, RADIX)
7  `m_member_policy_constructor(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define m_member_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
11     class POLICY` `_policy extends base_policy
12         typedef TYPE          l_field_array_t[];
13
14         local l_field_array_t  m_values;

```

```

15     local bit          l_exclude;
16     local string       m_radix=RADIX;
17
18     constraint c_policy_constraint {
19         (m_item != null) ->
20         ((l_exclude) ^ (m_item.FIELD inside {m_values}));
21     }
22
23     function new(
24         l_field_array_t values,
25         bit          exclude=1'b0,
26         string       radix=RADIX
27     );
28         this.set_values(values);
29         this.set_exclude(exclude);
30         this.set_radix(radix);
31     endfunction: new
32
33     virtual function string name();
34         return (`"POLICY`");
35     endfunction: name
36
37     virtual function string description();
38         string values_str = "";
39
40         foreach(m_values[i])
41             values_str = {
42                 values_str,
43                 $sformatf(m_radix, m_values[i]),
44                 i == m_values.size()-1 ? "" : ", ";
45
46         return ({
47             `"POLICY(FIELD ",
48             l_exclude ? "outside {" : "inside {" ,
49             values_str,
50             `"}`"
51         });
52     endfunction: description
53
54     virtual function policy copy();
55         copy = new(m_values, l_exclude, m_radix);
56     endfunction: copy
57
58     virtual function void set_values(l_field_array_t values);
59         this.m_values = values;
60     endfunction: set_values
61
62     virtual function l_field_array_t get_values();
63         l_field_array_t l_array;
64         foreach (this.m_values[i])
65             l_array[i] = this.m_values[i];
66         return (l_array);
67     endfunction: get_values
68

```

```

69     virtual function void set_exclude(bit exclude);
70         this.l_exclude = exclude;
71     endfunction: set_exclude
72
73     virtual function bit get_exclude();
74         return (this.l_exclude);
75     endfunction: get_exclude
76
77     virtual function void set_radix(string radix);
78         this.m_radix = radix;
79     endfunction: set_radix
80
81     virtual function string get_radix();
82         return (this.m_radix);
83     endfunction: get_radix
84 endclass: POLICY`*_policy
85
86 // Policy constructor definition
87 `define m_member_policy_constructor(POLICY, TYPE, RADIX="%0p")
88     typedef TYPE POLICY`*_array_t[];
89     static function POLICY`*_policy POLICY(
90         POLICY`*_array_t values,
91         bit exclude=1'b0,
92         string radix=RADIX
93     );
94         POLICY = new(values, exclude, radix);
95     endfunction: POLICY
96
97 `endif

```

E. range_policy.svh

```

1  `ifndef __RANGE_POLICY__
2  `define __RANGE_POLICY__
3
4  // Full policy definition
5  `define range_policy(POLICY, FIELD, TYPE, RADIX="%0p")
6  `m_range_policy_class(POLICY, FIELD, TYPE, RADIX)
7  `m_range_policy_constructor(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define m_range_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
11     class POLICY`*_policy extends base_policy
12         local TYPE l_low;
13         local TYPE l_high;
14         local bit l_exclude;
15         local string l_radix=RADIX;
16
17         constraint c_policy_constraint {
18             (m_item != null) -> (
19                 (l_exclude) ^
20                 ((m_item.FIELD >= l_low && m_item.FIELD <= l_high))
21             );
22         }

```

```

23
24     function new(
25         TYPE    low,
26         TYPE    high=low,
27         bit     exclude=1'b0,
28         string  radix=RADIX
29     );
30         this.set_range(low, high);
31         this.set_exclude(exclude);
32         this.set_radix(radix);
33     endfunction: new
34
35     virtual function string name();
36         return (`"POLICY`");
37     endfunction: name
38
39     virtual function string description();
40         return ({
41             ~(FIELD ",
42             l_exclude ? "outside [" : "inside [",
43             $sformatf(l_radix, l_low),
44             ", ",
45             $sformatf(l_radix, l_high),
46             ~"]`"
47         });
48     endfunction: description
49
50     virtual function policy copy();
51         copy = new(l_low, l_high, l_exclude, l_radix);
52     endfunction: copy
53
54     virtual function void set_range(TYPE low, TYPE high);
55         if (low <= high) begin
56             this.l_low = low;
57             this.l_high = high;
58         end else begin
59             this.l_low = high;
60             this.l_high = low;
61         end
62     endfunction: set_range
63
64     virtual function TYPE get_low();
65         return (this.l_low);
66     endfunction: get_low
67
68     virtual function TYPE get_high();
69         return (this.l_high);
70     endfunction: get_high
71
72     virtual function void set_exclude(bit exclude);
73         this.l_exclude = exclude;
74     endfunction: set_exclude
75
76     virtual function bit get_exclude();

```

```

77         return (this.l_exclude);
78     endfunction: get_exclude
79
80     virtual function void set_radix(string radix);
81         this.l_radix = radix;
82     endfunction: set_radix
83
84     virtual function string get_radix();
85         return (this.l_radix);
86     endfunction: get_radix
87 endclass: POLICY``_policy
88
89 // Policy constructor definition
90 `define m_range_policy_constructor(POLICY, TYPE, RADIX="%0p")
91     static function POLICY``_policy POLICY(
92         TYPE    low,
93         TYPE    high=low,
94         bit     exclude=1'b0,
95         string  radix=RADIX
96     );
97         POLICY = new(low, high, exclude, radix);
98     endfunction: POLICY
99
100 `endif

```